
Motivation

The problem of determining whether a system of constraints has a solution has

been studied in many areas of computer science and mathematics. In particular, de-

ciding if a system of inequalities has an integer solution is important because many

problems involve discrete values, such as loop indices, array positions and counters.

While the elimination algorithm Fourier-Motzkin has been implemented many

times to decide if a system has real solutions, implementations of the full Omega Test

are harder to find. The main reference remains the original algorithm introduced by

William Pugh [4], and in the context of formal verification, there exists a partial imple-

mentation in a system called Rocq [5], a proof assistant.

The use of Dafny is useful here because the goal of thesis is not only about com-

puting a result, but also about proving that each step is correct. Every time a variable

is eliminated or a new system is built, the implementation has to show that this new

system still has the expected relation with the original one.

For this reason, this thesis focuses on the implementation and formal verifica-

tion of the Omega Test algorithm in Dafny. In other words, this implementation of

the algorithm follows the mathematical steps taken to decide satisfiability, along with

specifications, postconditions, and lemmas that verify those steps.

Introduction

Linear arithmetic and algebra appear in many areas of computer science, such as

formal verification, program analysis and the optimization of compilers. Verification

problems can be solved by expressing the properties of a program as systems of linear

integer constraints and deciding their satisfiability.

One method that implements this verification of integer constraints is the Omega

Test proposed by William Pugh [4]. This method constructs three projections: Real

Shadow, which is used to decide unsatisfiability, Dark Shadow, used to detect integer

solutions and Grey Shadow, which searches for solutions outside of the Dark Shadow

and represents the completeness step of the algorithm.

This work represents an implementation of the Omega Test in Dafny [3], a pro-

gramming language used to add formal specifications to a program with the purpose

of verifying it. Those specifications consist of loop invariants (used to preserve prop-

erties during iterations), preconditions and postconditions (that describe the behavior

of a definition), and local assertions and lemmas (used to express different properties

and guide the verifier with intermediate steps).

0.1 A short presentation of Dafny specifications

A simple example that describes the syntax and purpose of Dafny is the next

method that computes the sum of the first n natural numbers:

method SumFirstN(n: nat) returns (sum: nat)
ensures sum == n * (n + 1) / 2

{
var i: nat := 0;
sum := 0;
while i < n

invariant 0 <= i <= n
invariant sum == i * (i + 1) / 2

{

3



i := i + 1;
sum := sum + i;
assert sum == i * (i + 1) / 2;

}
}

The first invariant describes the fact that the index is in the correct range, while

the second invariant verifies that the partial result is correct, meaning that the variable

sum stores the sum of the first i integers before and after each loop iteration. The

assertion is used to help the verifier establish this property locally. The final post-

condition, located in the method header, makes this property available to any call of

the method.

0.2 The Omega Test

The goal of the program is to prove that if the algorithm returns True, then there

exists an integer solution for the system, and if it returns False, no integer model sat-

isfies the system of inequalities. The first step in achieving this was to define data

structures that represent mathematical objects such as rational numbers, linear terms

and expressions, inequalities and systems, along with predicates and functions that

evaluate, compare and state specific properties about objects of those structures.

The implementation of the algorithm was then divided into small construction

steps, each one specified with the correct properties about the inequalities, systems

and the transformations applied to them. Those properties describe the consistency of

a constructed projection, the validity of a solution and the way satisfiability is decided

by the three shadows.

The following system exemplifies how the algorithm constructs each shadow and

decides satisfiability:

S =

 2 ≤ 2x,

3x ≤ 3

 Dark−−→ {0 ≤ −2} Real−−→ {1 ≤ x ≤ 1} Grey−−→


2 ≤ 2x,

3x ≤ 3,

2x = 2


The algorithm firstly constructs the Dark Shadow, and since this has no more

variables, it is checked for satisfiability and returns False. The Omega Test then gen-

erates the Real Shadow and finds it consistent (1 ≤ 1). Because the Real Shadow

is satisfiable and the Dark Shadow is not, the last step applied is the Grey Shadow,

which adds the equation 2x = 2 in the system and reruns the Omega Test.

Chapter 1

Mathematical Background

This chapter introduces the mathematical notions needed for the elimination pro-

cedures used in the implementation. It first presents linear expressions and systems of

inequalities, explaining how solutions are interpreted over real numbers and over inte-

gers. In the second part, it describes the main idea behind Fourier-Motzkin elimination

and why extending it to the Omega Test is necessary.

1.1 Linear Arithmetic

Linear arithmetic represents the mathematical foundation for introducing vari-

ables, linear expressions and systems of inequalities, along with operations over them.

A linear expression can be considered a sum of variables, each multiplied by a co-

efficient, together with a constant term. This way, we can write inequalities such as

a1x1 + a2x2 + · · · + anxn ≤ b, where variables represent unknown values and the coef-

ficients represent the relations between them.

A system of inequalities is a group of inequality statements that must all be true

at the same time. Each inequality represents an area in a geometrical space and the

solutions of a system are found in the area that lies at the intersection of all inequalities

of the system. If the variables can have real values, then the area of intersection can be

explored by algorithms such as Fourier-Motzkin. The problem becomes more difficult

when the solutions need to be integers, because finding non-null regions is not enough,

they also need to contain at least one point of integer coordinates.

These notions are used in this implementation to formalize concepts like linear

inequality systems and the operations performed to transform those systems. Each

constraint is interpreted as a relation between two linear terms, and variable elimina-

tion is done by forming new constraints, which are equivalent or approximate to the

ones in the original systems. Therefore, linear arithmetic is used as a theoretical base

to implement data structures and proofs in Dafny.

1.2 Fourier–Motzkin Elimination and the Omega Test

Fourier–Motzkin elimination [2] is a method for eliminating variables from sys-

tems of linear inequalities. It was originally introduced by Joseph Fourier and later re-

discovered by Theodore Motzkin and it works by combining lower and upper bounds

for a variable to produce a projection of the original system. This technique is mainly

used to reason whether a system of inequalities has real solutions or not.

Because we try to decide if the system has integer solutions and not just real

ones, simply projecting the Real Shadow is not enough: that is why we have to extend

the Fourier-Motzkin Elimination algorithm to the Omega Test proposed by William

Pugh [4].

This adaptation of variable elimination builds projections over integer constraints.

Unlike the real case, eliminating variables over integers requires additional checks,

because a non-empty real interval may contain no integer point. For this reason, the

Omega Test introduces constructions such as the Dark Shadow and the Grey Shadow.

It was developed as a practical algorithm for integer programming problems that

solves dependence analysis in compilers [1].

The compiler needs to determine whether two array accesses, possibly occurring

in different loop iterations, can refer to the same memory location. For example, it

may need to know whether a write access and a read access can target the same array

element. This question can be represented as a system of linear integer constraints,

the variables representing loop indices and the equalities expressing that the accessed

array positions are the same. If the resulting system has no integer solution, then there

is no dependence between the accesses, and the compiler may safely reorder or paral-

lelize the corresponding operations. Because array indices are integers, a real-valued

solution is not sufficient to prove the existence of an actual dependence.

Chapter 2

Formalization of Linear Inequalities

This chapter presents the datatypes needed for the implementation of the algo-

rithm, including rational numbers, linear expressions, inequalities, and systems of in-

equalities. It also introduces the functions and operations used to manipulate and

evaluate these structures and also the predicates used to describe properties such as

validity or equivalence.

2.1 Rational Numbers

The custom datatype Rational, shown in code snippet 2.1, has three fields which

represent a fraction and its mathematical value: a numerator num, a positive and non-

zero denominator den and the mathematical value of the fraction val. The denomina-

tor is expressed through a type called pos, defined as a strictly positive integer. Ex-

pressing the denominator this way instead of using a predicate that requires the value

to be strictly greater than 0 significantly helps simplify the verification. When defining

a subset type in Dafny, the verifier needs to know the type is not empty, therefore a

witness is provided, in this case, the positive integer 1.

The relation between the fields is expressed through the predicate validRational

to be num
den

= val (as it can be seen in the code snippet below). This predicate is present

in every operation over the objects of the algorithm and needs to be true at all times in

the execution, otherwise the methods could end up working with inconsistent values,

such as Rational(1, 2, 7.0), which is mathematically incorrect.

type pos = n: int | n > 0 witness 1
datatype Rational = Rational(num: int, den: pos, val: real)
ghost predicate validRational(r: Rational)
{

r.val == r.num as real / r.den as real
}

Code snippet 2.1: Definition of the Rational datatype

The module rational also defines different operations such as addition, subtraction,

multiplication or conversion from int to Rational. For example, the addition function

(shown in code snippet 2.2) requires the parameters to be valid rationals and ensures

that the output is valid and its result is consistent with the addition by using the val

field. For a
b

and c
d
, the result will be ad+cb

bd
. The subtraction function is implemented in

a similar way.

function add(rat1 : Rational, rat2 : Rational) : Rational
requires validRational(rat1)
requires validRational(rat2)
ensures validRational(add(rat1, rat2))
ensures add(rat1, rat2).val == rat1.val + rat2.val

{
var den’ := rat1.den * rat2.den;
var num’ := rat1.num * rat2.den - rat2.num * rat1.den;
Rational(num’, den’, num’ as real / den’ as real)

}

Code snippet 2.2: Definition of the function subtract

Multiplication between two Rational values is implemented by multiplying the

corresponding numerators and denominators. The resulting rational number has nu-

merator rat1.num * rat2.num and denominator rat1.den * rat2.den. Since

both denominators are strictly positive, their product is also strictly positive, so the

result satisfies the denominator requirement of the Rational datatype.

function multiply(rat1 : Rational, rat2 : Rational) : Rational
requires validRational(rat1)
requires validRational(rat2)
ensures validRational(multiply(rat1, rat2))
ensures multiply(rat1, rat2).val == rat1.val * rat2.val

{
var den’ := rat1.den * rat2.den;
var num’ := rat1.num * rat2.num;
Rational(num’, den’, num’ as real / den’ as real)

}

Code snippet 2.3: Definition of the function multiply

There are a few lemmas in this module that help prove basic arithmetic concepts

like reflexivity, symmetry, transitivity, commutativity, associativity and distributivity,

needed to verify future operations on linear expressions. For example, the lemma Mul-

SumSimple (shown in code snippet 2.4) proves by induction the property of distribu-

tivity over sequences:
∑

i(si ∗ factor) = (
∑

i si) ∗ factor . Since the body of the lemma

implements induction explicitly, we have to mention to Dafny to disable automatic in-

duction, otherwise it might not be able to prove the properties. We also need to provide

a decreases clause, since the lemma recursively calls itself and the verifier needs a

guarantee that the proof will eventually end. In this case, the measure is the length

of s1 because at each recursive call, the lemma is applied to the tail s1[1..], whose

length is strictly smaller than the original sequence.

lemma {:induction false} MulSumSimple(s1: seq<Rational>,
s2: seq<Rational>, factor: Rational)

requires validRational(factor)
requires forall i :: 0 <= i < |s1| ==>

validRational(s1[i])
requires forall i :: 0 <= i < |s2| ==>

validRational(s2[i])
requires |s1| == |s2|
requires forall i :: 0 <= i < |s2| ==> s2[i] ==

multiply(s1[i], factor)
ensures SumSeq(s2).val == multiply(SumSeq(s1), factor).val
decreases |s1|

Code snippet 2.4: Definition of the lemma MulSumSimple

2.2 Linear Expressions and Inequalities

The module dataTypes uses the Rational type to define Terms, Linear Expres-

sions and Inequalities (shown in code snippet 2.5). A Term has two fields, a name

and a Rational coefficient, so it can be used to express objects such as 2x. A Linear-

Expr consists of a sequence of terms and a Rational constant: 2x + 4y − z + 2. An

Inequality has two LinearExpr fields, representing the left and the right sides of an

inequality: 2x+ 4y − z + 2□− x+ 4y − 1.

datatype Term = Term(name: string, coeff: Rational)
datatype LinearExpr = LinearExpr(terms: seq<Term>,

constant: Rational)
datatype Inequality = Inequality(left: LinearExpr,

right: LinearExpr)

Code snippet 2.5: Definitions of different algebraic types

These types define only the mathematical structure of the constraints and not

their direct evaluation. For example, LinearExpr does not actually store the sum of

some terms, only the sequence of terms and the constant. Similarly, the Inequality

only represent each side without applying any relational operator like ≤ or >. This

separation and symbolic representation is necessary because the algorithm works not

only with numerical values, but also with different structures that contain variables

and that need to be modified during different processes. For example, terms need

to be moved around, summed, eliminated, which would be difficult if a LinearExpr

would be only expressed by a function.

The evaluation of those types is done through different functions or predicates

that are defined in the module evaluating (a few examples are shown in code snip-

pet 2.6). This is done incrementally, by first evaluating a Term.

• The type Valuation associates an integer value to a string. This is the way we can

attribute values to variables.

• The function EvalTerm takes as input a Term and a Valuation and simply mul-

tiplies the coefficient with the value of the variable, mapped by the Valuation.

However, because the variable values are of type int and the multiply function

needs Rational parameters, we first use IntToRational to create a Rational object.

• Evaluating LinearExpr is done by EvalExpr, which uses an additional function

that sums a sequence of terms, then adds the result to the constant.

• Finally, there are more ways to evaluate an Inequality. The one used the most in

this implementation is EvalInequality that subtracts the right side from the left,

then applies the relational operator ≤, comparing the result to 0. In other words,

it performs left − right ≤ 0.0. Another function, used in some intermediate proof,

is EvalInverseStrictInequality, which performs left − right > 0.0.

type Valuation = string -> int
function EvalTerm(t: Term, val: Valuation): Rational

requires validRational(t.coeff)
ensures validRational(EvalTerm(t, val))

{
multiply(t.coeff, IntToRational(val(t.name)))

}
function EvalExpr(e: LinearExpr, val: Valuation): Rational

requires ValidTerms(e.terms)
requires validRational(e.constant)

{
add(SumTerms(e.terms, val), e.constant)

}
function EvalInequality(i: Inequality, val: Valuation): bool

requires JustValidInequality(i)
{

var diff := subtract(EvalExpr(i.left, val),
EvalExpr(i.right, val));
diff.val <= 0.0

}

Code snippet 2.6: Evaluating structures

Each function that processes Rational objects requires and ensures that the values

are valid. This is done with different predicates (shown in code snippet 2.7) that build

on validRational (from snippet 2.1), describing requirements for sequences of terms,

linear expressions or inequalities. For example, a valid sequence of terms needs to have

all coefficients valid Rationals. An inequality is valid if all its terms and constants are

valid, and additionally, all variables are unique.

ghost predicate ValidTerms(ts: seq<Term>)
{

forall i :: 0 <= i < |ts| ==> validRational(ts[i].coeff)
}
ghost predicate ValidInequality(i: Inequality)
{

ValidTerms(i.left.terms) &&
ValidTerms(i.right.terms) &&
validRational(i.left.constant) &&
validRational(i.right.constant) &&
UniqueTerms(i.left.terms) && UniqueTerms(i.right.terms)

}
ghost predicate ValidSystem(s: IneqSystem)
{

forall i :: 0 <= i < |s.ineqs| ==>
NormalizedInequality(s.ineqs[i])

}

Code snippet 2.7: Valid structures

Because objects go through processes like normalization or other transformations,

we need to ensure that two syntactically different structures have the same semantic

meaning. This is done by using the predicates that build on RationalEq, which checks

if two fractions have the same mathematical value. A few examples can be seen in the

code snippet 2.8.

• EquivalentTerms checks if for all valuations, two sequences of terms have the

same sum. For example, those two sequences are considered equivalent:

3x,−1y, 2x, 4y and 3y, 5x.

• EquivalentExpressions compares the evaluation of two LinearExpr using Ratio-

nalEq.

• EquivalentInequalities directly compares the evaluation of two Inequalities.

ghost predicate EquivalentTerms(ts1: seq<Term>, ts2:
seq<Term>)
requires ValidTerms(ts1)
requires ValidTerms(ts2)

{
forall val: Valuation :: RationalEq(SumTerms(ts1, val),

SumTerms(ts2, val))
}

ghost predicate EquivalentExpressions(e1: LinearExpr, e2:
LinearExpr)
requires ValidTerms(e1.terms)
requires ValidTerms(e2.terms)
requires validRational(e1.constant) &&

validRational(e2.constant)
{

forall val: Valuation :: RationalEq(EvalExpr(e1, val),
EvalExpr(e2, val))

}

Code snippet 2.8: Equivalent structures

2.3 Systems of inequalities

A group of more Inequalities are represented in a system. The simplest form is

the structure IneqSystem (presented in code snippet 2.9), which simply stores a se-

quence of Inequalities. Its semantic meaning is expressed separately by the predicate

SystemHolds, which checks if all inequalities hold under a certain Valuation. Because

systems go through different transformations, auxiliary predicates are needed, such as

EquivalentSystems that only checks if two systems hold under the same Valuations,

or NormalizedSystem. For a system to be normalized, its Inequalities have to be of

the form a1x1 + ... + anxn ≤ c, with no repeating variables. A sequence of Terms is

checked by UniqueTerms not to have repeating variables (so term.name).

Another important type of system is the PartitionedSystem (shown in code snip-

pet 2.9), which is used to categorize inequalities by the coefficient of a certain variable.

It has three fields, all of the form seq<Inequality>:

• pos: inequalities of the form ax ≤ b1y1 + ...+ bnyn + c;

• neg: inequalities of the form b1y1 + ...+ bnyn + c ≤ ax;

• zero: inequalities of the form b1y1 + ... + bnyn + c ≤ 0.0, that do not contain x at

all.

This structure is mostly used in the process of eliminating a certain variable and

constructing different projections of systems. It has its own predicates, such as Ter-

mInPartSystem, that looks for a certain variable, PartitionedSystemHolds that checks

all evaluations of the Inequalities stored in its fields, and ValidPartitionedSystem that

states that all inequalities contained in the positive, negative, and zero partitions are

valid and can therefore be safely used in later verification steps:

ghost predicate TermInPartSystem(ps: PartitionedSystem, x:
string)

{
(exists i :: 0 <= i < |ps.pos| && TermInIneq(ps.pos[i],

x)) ||
(exists i :: 0 <= i < |ps.neg| && TermInIneq(ps.neg[i],

x)) ||
(exists i :: 0 <= i < |ps.zero| && TermInIneq(ps.zero[i],

x))
}
ghost predicate PartitionedSystemHolds(ps: PartitionedSystem,

val: Valuation)
requires ValidPartitionedSystem(ps)

{ (forall i :: 0 <= i < |ps.pos| ==>
EvalInequality(ps.pos[i], val)) &&
(forall i :: 0 <= i < |ps.neg| ==>

EvalInequality(ps.neg[i], val)) &&
(forall i :: 0 <= i < |ps.zero| ==>

EvalInequality(ps.zero[i], val))
}
ghost predicate ValidPartitionedSystem(ps: PartitionedSystem)
{

(forall i :: 0 <= i < |ps.pos| ==>
ValidInequality(ps.pos[i])) &&
(forall i :: 0 <= i < |ps.neg| ==>

ValidInequality(ps.neg[i])) &&
(forall i :: 0 <= i < |ps.zero| ==>

ValidInequality(ps.zero[i]))
}

In the construction of the Dark Shadow, a new relation needs to be expressed: a

lower bound L ≤ ax and an upper bound bx ≤ U are used to form a new constraint

bL − aU ≤ b + a − ab − 1. This relation between a pair and a new constraint is stored

using the structures PairDark and DarkTraceSystem (shown in code snippet 2.9) (for

a whole system). This way, when forming the dark projection, we can keep track of

the inequalities that formed the new constraints and prove the soundness of the con-

struction. Some predicates specific to this type of system are PairDarkTracksSplit(p,

ps), which says: p.a in ps.pos && p.b in ps.neg, and PairDarkTracksGener-

ated(p, g, ps), which connects the generated inequality to its source pair. To transform

this trace in a normal system, we use the functions like TraceToIneqSystem or Materi-

alizeDarkTrace.

datatype IneqSystem = IneqSystem(ineqs: seq<Inequality>)
datatype PartitionedSystem = PartitionedSystem(

pos: seq<Inequality>,
neg: seq<Inequality>,
zero: seq<Inequality>)

datatype PairDark<A, B, C> = PairDark(a: A, b: B, c: C)
datatype DarkTraceSystem = DarkTraceSystem(

base: seq<Inequality>,
pairs: seq<PairDark<Inequality, Inequality, Inequality>>)

Code snippet 2.9: System structures

There are also systems associated to the Grey Shadow equations, but they do

not appear as separate datatypes. A grey branch is built as an IneqSystem, for exam-

ple using GreyBranchSourceSystemFromALI(sys, a, L, x, i), which adds to a source

system a new constraint of type GreyEquality(L, i, a, x, val) (an equality of the form

a ∗ x = L+ i).

Another important predicate, used in deciding final satisfiability, is Variable-

FreeSystem, which checks that the right and left side of all its inequalities do not con-

tain any terms, in other words: |i.ineq.left.terms| == 0.0 &&

|i.ineq.right.terms| == 0.0, for all inequalities in the system. In a system of

this form, the satisfiability does not depend anymore on variables and can be decided

by simply checking if those constant inequalities are true.

ghost predicate VariableFreeSystem(sys: IneqSystem)
{

forall i :: 0 <= i < |sys.ineqs| ==>
|sys.ineqs[i].left.terms| == 0 &&

|sys.ineqs[i].right.terms| == 0
}

A few other important predicates, used in the correctness proof of the Dark

Shadow are ones that check if all coefficients and constants of an object are integers (a

few examples are shown in code snippet 2.10). Since we work with Rationals, check-

ing this implies verifying that rational .den == 1. Because the values variables can take

are integers and we start with integer coefficients and only multiply, add, subtract or

negate (division appears only in intermediate proof), every step of constructing the

Dark Shadow also works with integers.

predicate IntegerCoeffSeq(ts: seq<Term>)
{

forall i :: 0 <= i < |ts| ==> ts[i].coeff.den == 1
}
predicate IntegerCoeffIneq(ineq: Inequality)
{

IntegerCoeffSeq(ineq.left.terms) &&
IntegerCoeffSeq(ineq.right.terms)
&& ineq.left.constant.den == 1 && ineq.right.constant.den

== 1
}
predicate IntegerCoeffSys(sys: IneqSystem)
{

forall i :: 0 <= i < |sys.ineqs| ==>
IntegerCoeffIneq(sys.ineqs[i])

}

Code snippet 2.10: Integer structures

Chapter 3

Verified Fourier–Motzkin Elimination

Figure 3.1: Control flow of the Omega Test

OmegaTest is the main method that decides the satisfiability of a system by con-

structing the Real, Dark and Grey shadows. Each of them has a different role:

• the Real Shadow is an over-approximation

• the Dark Shadow is an under-approximation

• the Grey Shadow is a search for a solution that is in the Real projection, but not

in the Dark projection.

The usual Omega Test flow is Real Shadow → Dark Shadow → Grey Shadow.

The reason behind choosing to do the Dark projection first is that this shadow offers

a sure path to decide satisfiability. If the recursion on the Dark Shadow returns True,

the algorithm can immediately return also True [4].

If, however, the Dark Shadow returns False, we cannot decide on the unsatis-

fiability for the original system. On this path, the algorithm next constructs the Real

Shadow. Since this projection is an over-approximation, if the Real Shadow returns an

unsatisfiable system, then the algorithm can return False, deciding the original system

is unsatisfiable.

The final case is if the Dark Shadow returns False and the Real Shadow returns

True. Then there might still exist a solution in the space between those projections, in

an interval smaller than one unit.

Therefore, the algorithm builds a limited set of equations of the form a*x = L +

i, where L is a lower bound in the system for the variable currently being eliminated

and i takes the values of all integers in a finite interval. By adding this equality to the

system, we form a Grey Shadow on which we recall the Omega Test. If any of these

recursive calls returns True, then the original system has an integer solution; if all of

them fail, then no solution exists in the grey region considered by the algorithm and

we return False.

3.1 Real Shadow

The Real Shadow is the over-approximation of the system, represented by a pro-

jection over real values: any integer solution that satisfies the original system is also a

real solution of the projection. Due to this fact, we can use the Real Shadow to decide

unsatisfiability: Real Shadow UNSAT =⇒ System UNSAT.

3.1.1 Construction of Projected Constraints

The main method that builds this projection is RealShadow. It takes as input a

system and an elimination order and returns a variable-free system, which represents

the final Real projection of the original system.

It has three important steps (presented in code snippet 3.1): choosing the variable

that will be eliminated, building the projection and normalizing it.

method RealShadow(sys: IneqSystem, elimOrder: seq<string>)
returns (reduced: IneqSystem)

requires ValidSystem(sys)
requires forall x :: TermInSystem(sys, x) ==> x in

elimOrder
requires forall i, j :: 0 <= i < j < |elimOrder| ==>

elimOrder[i] != elimOrder[j]
ensures ValidSystem(reduced)
ensures forall val :: SystemHolds(sys, val) ==>

SystemHolds(reduced, val)
ensures (forall val :: !SystemHolds(reduced, val)) ==>

(forall val :: !SystemHolds(sys, val))
ensures forall y :: y in elimOrder ==>

!TermInSystem(reduced, y)
ensures VariableFreeSystem(reduced)

{
reduced := sys;
var i := 0;

while i < |elimOrder| {
var x := elimOrder[i];
var proj := CombineIneqs(reduced, x);
var newReduced := NormalizeSystem(proj);
reduced := newReduced;
i := i+1;

}
}

Code snippet 3.1: The main structure of the RealShadow method

The first precondition required by the method is for the system to be valid, mean-

ing each inequality needs to be normalized. The sequence elimOrder needs to contain

every term in the system and also to not repeat variables in its list. Using those require-

ments, the method returns a projection that is guaranteed by the post-conditions to be

valid, to be variable-free and to be satisfied by any model that satisfies the original

system.

CombineIneqs is the method that eliminates the variable x, using several other

definitions:

CombineIneqs

SplitMembersHoldUnderValutation

SplitSystem

PartitionSingleIneq

Sum Coeff FilterNotX PartitionPosIneq PartitionPosIneq PartitionNegIneq

BuildProjectedPairs

CombinePair

DivideIneqByXCoeff NegateTermSeq Normalize RemoveZeroTerms

Figure 3.2: Methods used by CombineIneqs

The method CombineIneqs (presented in the figure 3.2) receives a normalized

system and firstly calls a method to transform the IneqSystem into a PartitionedSys-

tem (shown in code snippet 2.9). This is done mainly by the method SplitSystem,

which iterates through each inequality and categorizes it in the three different fields of

the variable ps (which is an object of type PartitionedSystem), by the coefficient a of

the variable x:

• if a > 0, then the inequality is added to ps.pos;

• if a < 0, then it is added to ps.neg;

• if a == 0, or the variable x does not appear at all, the inequality is added to

ps.zero.

In order to make the categorization of a single inequality simpler, CombineIneqs calls

PartitionSingleIneq. This method receives an Inequality and processes it by separat-

ing the terms in ”contains x” and ”does not contain x” (by using a function FilterNotX),

then calling other methods that process the structure of the inequality to fit the neces-

sary post-conditions. A few important post-conditions are:

• inequalities in ps.neg are of the form L.terms + L.constant ≤ ax + 0.0, where

L.terms does not contain x and a.val > 0.0;

• inequalities in ps.pos are of form bx+0.0 ≤ U .terms+U .constant , where U .terms

does not contain x and b.val > 0.0;

• inequalities in ps.zero do not contain x and are normalized;

• for each inequality p in ps, there exists an inequality in the original system, i, such

that EquivalentInequalities(i, p).

The most important post-condition is the last one, since keeping track of system

equivalence is crucial in proving satisfiability. The proof starts from one of the methods

that change the shape of an inequality, for example PartitionPosIneq (presented in the

code snippet 3.2 from below), which isolates the variable on the left.

var x_term := FormTerm(coeff_x, x);
var neg_other_terms := NegateTermSeq(other_terms);

res := Inequality(
LinearExpr([x_term], Rational(0, 1, 0.0)),
LinearExpr(neg_other_terms, ineq.right.constant));

Code snippet 3.2: PartitionPosIneq

Proving that a∗x+ rest ≤ c is equivalent to a∗x ≤ c− rest is firstly done with the

help of the lemma SumTermsSplit, which shows that the sum of the initial terms can

be split in terms that contain x and terms that do not contain x. After this, Partition-

IneqPreservesDiff proves that moving some terms from one part of the inequality to

the other keeps the truth value of the inequality.

Another important aspect that needs to be proved is the variable x does not ap-

pear anymore on the side it was moved from. This is done by the lemmas NotXSe-

qProp and NegateTermSeqPreservesNames, which prove that x is not introduced af-

ter filtering or negating terms.

The post-conditions of the methods that process one inequality contain those

properties, but SplitSystem also has to keep track of them in the while loop under the

form of invariants or asserts. Because the algorithm works very often with sequences,

a pattern of proof frequently used is: during a step of a loop we assert properties about

all the elements we already processed and about the element we are currently process-

ing (as can be seen in the code snippet 3.3).

if j < |prevPosIneqs| {
assert posIneqs[j] == prevPosIneqs[j];

} else {
assert j == |prevPosIneqs|;
assert posIneqs[j] == part;

}

Code snippet 3.3: Example of proof pattern

After obtaining the PartitionedSystem, the method CombineIneqs calls Build-

ProjectedPairs, which uses two while loops to iterate through each pair of (posi-

tive inequality, negative inequality) to form a new inequality. It chooses a pair p

and n, asserts that they have the correct form, and calls CombinePair(p, n, x),

which is responsible with building the new constraint from the pair. It firstly uses Di-

videIneqByXCoeff, which receives three parameters: an inequality ineq , a variable x

and a boolean value called left. The purpose of left is to indicate on which side of the

inequality appears x:

• if left is true, then ineq is a pos inequality, meaning x is on its left side;

• if left is false, then ineq is a neg inequality, meaning x is on its right side.

Depending on the value of left , the method DivideIneqByXCoeff takes the co-

efficient of the Term containing x, computes its reciprocal and then uses a method to

scale the inequality with the reciprocal. It uses different lemmas to assert properties

about the resulting inequality: that after scaling, the length of the terms sequence does

not change, that x stays on its side of the inequality, that no new variable is introduced

and the uniqueness of the terms is preserved, and finally that the two inequalities (the

one received and the returned one) are equivalent.

After dividing both inequalities p and n by the specific coefficient of x, Combine-

Pair builds each side of the new, already normalized inequality:

• the terms on the left side are computed

n.left.term + NegateTermSeq(p.right.terms);

• similarly for the constant on the right:

add(p.right.constant, multiply(Rational(-1, 1, -1.0),

n.left.constant));

• the left constant and the right sequence of terms are just left empty:

res := Inequality(
LinearExpr(leftFinal, Rational(0, 1, 0.0)),
LinearExpr([], rightConst));

The mathematical transformation applied here is represented by starting from a

pair of inequalities L+ c1 ≤ ax,

bx ≤ U + c2,

where a, b > 0. We first divide each inequality by the coefficient of x:
L

a
+

c1
a

≤ x,

x ≤ U

b
+

c2
b
.

Therefore, the lower bound must be smaller than or equal to the upper bound:

L

a
+

c1
a

≤ U

b
+

c2
b
.

The algorithm directly builds the normalized version:

L

a
− U

b
≤ c2

b
− c1

a
.

This way, the variable x is eliminated from the two bounds. Repeating this for

every pair of lower bound and upper bound eliminates x from the whole system. The

method BuildProjectedPairs keeps track of this important property using invariants

and the post-conditions from CombinePair:

ensures forall j :: 0 <= j < |res.left.terms| ==>
res.left.terms[j].name != x

ensures forall j :: 0 <= j < |res.right.terms| ==>
res.right.terms[j].name != x

Code snippet 3.4: CombinePair post-conditions

At the end of CombineIneqs, the method can declare in its post-conditions that:

!TermInSystem(proj, x), returning to RealShadow (with the structure presented

in code snippet 3.1) the real projection that eliminated the variable x. The next step is

to normalize the new system by calling NormalizeSystem(proj), which ensures:

ValidSystem(norm).

By repeating this procedure and iterating through elimOrder (as can be seen in

the code snippet 3.1), the algorithm eliminates one variable at a time. After each step,

the resulting normalized projection becomes the new current system, on which the

next elimination step is applied.

22



The final system that RealShadow returns is a variable-free system, with each

inequality of the form c1 ≤ c2, or more specific to the normalization: 0 ≤ c.

A small example that illustrates the Real projection:
2x+ y ≤ 8,

−3x+ 2y ≤ 3,

−y ≤ −1


eliminate x−−−−−−→

 y ≤ 30

7
,

−y ≤ −1

 eliminate y−−−−−−→
{
0 ≤ 138

42

}
.

3.1.2 Soundness and Completeness

The main property of the Real Shadow is SAT(S) =⇒ SAT(RealShadow(S)).

This is illustrated in the method by the post-condition: forall val ::

SystemHolds(sys, val) =⇒ SystemHolds(reduced, val). Since we use

the Real Shadow to prove unsatisfiability, another post-condition, which results from

the first one, was added to the method to facilitate further proof:

(forall val :: !SystemHolds(reduced, val)) =⇒

(forall val :: !SystemHolds(sys, val)).

The reason behind the fact that the real projection by itself cannot establish sat-

isfiability is that we cannot guarantee the existence of an integer value in an interval

bounded by two real values. For example, if for a certain valuation of the other vari-

ables, we get 1.3 ≤ x ≤ 1.8, the projection will be 1.3 ≤ 1.8, which is true, and therefore

is satisfiable. However, the system from which the projection originated is not satisfi-

able, since there is no integer value between 1.3 and 1.8.

The proof of the two main post-conditions in the method RealShadow is done in

two main steps:

• a local proof in CombineIneqs for one elimination step;

• a global proof in RealShadow for all variables in elimOrder.

In CombineIneqs, the main soundness post-condition says that for any valuation

of variables that satisfies the original system, the real projection is also satisfied:

ensures forall val :: SystemHolds(sys, val) ==>
SystemHolds(proj, val)

The proof is built on the previous steps. For example, SplitMembersHoldUn-

derValuation was added as an intermediate step between CombineIneqs and Split-

System with the only purpose of proving the implication between the system and its

partitioned form: ensures forall val :: SystemHolds(sys, val) =⇒

PartitionedSystemHolds(ps, val).

We start from CombinePair, the method that builds one new inequality start-

ing from a lower bound and an upper bound. Each step applied to the inequalities is

stored in a separate variable. In the final proof block, these variables can then be used

to formally verify the transformations step by step. For example, after constructing

the left side of the new constraint, we prove that it represents the difference between

the terms from the upper bound and the terms from the lower bound. Following each

transformation using asserts, we can finally prove that the Valuation that satisfies the

two bounds also satisfies the new constraint, res.

forall val | EvalInequality(posIneq, val) &&
EvalInequality(negIneq, val)

ensures EvalInequality(res, val)
{

assert EvalInequality(posUnit, val);
assert EvalInequality(negUnit, val);
...
assert SumTerms(left’, val).val ==

SumTerms(negUnit.left.terms, val).val -
SumTerms(posUnit.right.terms, val).val;

...
assert SumTerms(left’, val).val <= rightConst.val;
...
assert EvalExpr(res.left, val).val <=

EvalExpr(res.right, val).val;
assert EvalInequality(res, val);

}

The final post-condition in CombinePair is:

ensures forall val :: EvalInequality(posIneq, val) &&

EvalInequality(negIneq, val) =⇒ EvalInequality(res, val).

Using this statement, we can prove the implication between the partitioned form

of the system and its projection. The idea behind lifting the proof from a pair of bounds

to the whole system is that each time a new constraint is build, we prove its sat-

isfiability. We know the partitioned system holds, which means each pair of lower

bound x upper bound holds. We then use the post-condition of CombinePair to show

that the resulting inequality also holds. Each new inequality is added to a sequence

projected, and the final returned system is built by also adding the inequalities

from ps.zero (the ones that do not contain x): proj := IneqSystem(projected

+ ps.zero).

Therefore, each inequality in the projection either comes from a pair of bounds,

which we proved to be satisfiable, either comes from ps.zero. So we split on cases and

prove that when the partitioned system holds, the projection also holds.

assert forall k :: 0 <= k < |proj.ineqs| ==>
EvalInequality(proj.ineqs[k], val) by {
forall k | 0 <= k < |proj.ineqs|

ensures EvalInequality(proj.ineqs[k], val)
{

if k < |projected| {
assert proj.ineqs[k] == projected[k];

} else {
var idx := k - |projected|;
assert 0 <= idx < |ps.zero|;
assert proj.ineqs[k] == ps.zero[idx];
assert EvalInequality(ps.zero[idx], val);

}
}}

The local completeness proof is derived from the soundness proof. Suppose val

does not satisfy the projection. If val satisfies the original system, then it is implied that

it has to also satisfy the projection. But this contradicts the initial assumption, therefore

val cannot satisfy the original system.

assert forall val :: !SystemHolds(proj, val) ==>
!SystemHolds(sys, val) by {

forall val | !SystemHolds(proj, val)
ensures !SystemHolds(sys, val)

{
if SystemHolds(sys, val) {

assert SystemHolds(proj, val); }
}}

The global proof is implemented in RealShadow (with the structure presented

in the code snippet 3.1). This method keeps track in its while loop of the properties

involving each projection built iteratively:

invariant 0 <= i <= |elimOrder|
invariant ValidSystem(reduced)
invariant forall val :: SystemHolds(sys, val) ==>

SystemHolds(reduced, val)
invariant (forall val :: !SystemHolds(reduced, val)) ==>

(forall val :: !SystemHolds(sys, val))
invariant forall x :: x in elimOrder[..i] ==>

!TermInSystem(reduced, x)
invariant forall y :: TermInSystem(reduced, y) ==> y in

elimOrder

Code snippet 3.5: All the invariants in RealShadow

The chain of proof that applies to keep track of the satisfiability properties is:

SystemHolds(sys, val)
=> SystemHolds(reduced, val) //using invariant
=> SystemHolds(proj, val) //using CombineIneqs
=> SystemHolds(newReduced, val) //using EquivalentSystems

The last postcondition of RealShadow is (ensures

VariableFreeSystem(reduced), which says that the final projection has no more

variables. This property is proved by the last two invariants (shown in code snip-

pet 3.5). At each step, we know that all variables already eliminated (in elimOrder[..i])

do not appear anymore in the system and also that all variables in the system are

in elimOrder, therefore we can prove that at the end of the loop every variable in

elimOrder has been eliminated. Since there is no variable in reduced that is not also

in elimOrder, we can conclude that there are no more variables in reduced.

3.2 Dark Shadow

The Dark Shadow is an under-approximation of the system, represented by a

projection over integer values which keeps only the regions wide enough to fit an in-

teger. This is why, if the dark projection has a solution, the original system also has a

solution. Therefore, we use the Dark Shadow to decide satisfiability: Dark Shadow

SAT =⇒ System SAT.

3.2.1 Construction of Projected Constraints

The main method of this section is DarkShadow. Its implementation is similar to

the implementation of the real projection (shown in code snippet 3.1): it takes as input

a system and an elimination order and returns a variable-free system, representing the

Dark projection of the original system.

This method can be used on its own, just like RealShadow, this time to prove

the satisfiability of a system, containing post-conditions that confirm that any model

that satisfies the returned projection also satisfies the original system. The method

also guarantees that if it receives a valid system that only contains terms with inte-

ger coefficients, it returns a valid, variable-free system that also contains only integer

coefficients.

method DarkShadow(sys: IneqSystem, elimOrder: seq<string>)
returns (reduced: IneqSystem)

requires ValidSystem(sys)
requires IntegerCoeffSys(sys)
requires forall x :: TermInSystem(sys, x) ==> x in

elimOrder
requires forall i, j :: 0 <= i < j < |elimOrder| ==>

elimOrder[i] != elimOrder[j]
ensures ValidSystem(reduced)
ensures forall y :: y in elimOrder ==>

!TermInSystem(reduced, y)
ensures VariableFreeSystem(reduced)
ensures IntegerCoeffSys(reduced)
ensures (exists v :: SystemHolds(reduced, v)) ==> (exists

w :: SystemHolds(sys, w))

Code snippet 3.6: The definition of the DarkShadow

However, OmegaTest (presented in the figure 3.1) does not use this method di-

rectly, and only uses its steps. The reason behind this is that after eliminating each

variable, if the projection is unsatisfiable, we need to check the real projection and

eventually the grey ones, so we need the projection that was proven unsatisfiable and

not the variable-free one.

One step of Dark Shadow in the Omega Test is represented by the methods:

var darkTrace, ps := DarkEliminateStepTrace(sys, x);
var darkStep := MaterializeDarkTrace(darkTrace);
var darkValid := OmegaTest(darkStep, nextElimOrder);

Out of those, the one method that builds the actual steps for the elimination is

DarkEliminateStepTrace. This method first calls SplitSystemForDark, which, simi-

larly to SplitSystem, transforms an IneqSystem to a PartitionedSystem.

DarkEliminateStepTrace

SplitSystemForDark

PartitionSingleIneq

Sum Coeff FilterNotX PartitionPosIneq PartitionPosIneq PartitionNegIneq

ExistsXInTrace

PartitionDark

BuildGeneratedPair

GetBoundsAndPair DarkStep BuildBPlusAMinusAB BuildNormalizedConstraint

Figure 3.3: Methods used by DarkEliminateStepTrace

The difference between SplitSystemForDark and the partitioning method from

the partitioning method from Real Shadow is that the dark one adds stronger post-

conditions:

ensures forall val :: SystemHolds(s, val) ==>
PartitionedSystemHolds(ps, val)

ensures forall i :: 0 <= i < |s.ineqs| ==>
(exists pp :: pp in ps.pos &&

EquivalentInequalities(s.ineqs[i], pp)) ||
(exists nn :: nn in ps.neg &&

EquivalentInequalities(s.ineqs[i], nn)) ||
(exists zz :: zz in ps.zero &&

EquivalentInequalities(s.ineqs[i], zz))

Code snippet 3.7: SplitSystemForDark post-conditions

Otherwise, this method returns a system split depending on the coefficient of a

given variable x. It constructs it in the same manner and with the same methods as

SplitSystem.

The method ExistsXInTrace acts as an intermediate between PartitionDark and

DarkEliminateStepTrace and it is used to prove satisfiability.

The method PartitionDark is used to transform a PartitionedSystem into a Dark-

TraceSystem (with the definition presented in code snippet 2.9). This is a new type of

system structure, introduced to facilitate the proof of the Dark Shadow. The first field

of this structure, base, simply stores all the inequalities from ps.zero, the ones that do

not contain x. The second field, pairs, memorizes each pair of upper × lower bounds

and the new constraint they built. To memorize one of those relations, it uses the

datatype PairDark: the triplet PairDark(upperIneq, lowerIneq, darkIneq)

stores p.a = upperIneq, p.b = lowerIneq, p.c = darkIneq.

The new constraint p.c is build using the method BuildGeneratedPair, which

starts from two bounds of the type L ≤ ax and bx ≤ U and builds a new inequality

a ∗ U − b ∗ L > a ∗ b− a− b.

The method starts by building the four important components, by calling Get-

BoundsAndPair:

lowerBound = L
upperBound = U
a = the coefficient of x in the lower bound
b = the coefficient of x in the upper bound

After this, the method builds the left side of the new constraint by calling Dark-

Step, resulting in a∗U−b∗L. The limit on the right is built using BuildBPlusAMinusAB.

The final step is building the normalized inequality using those two parts. Because in-

equalities like U > L only appear in intermediate proof and systems do not have the

option of evaluating such expressions, the constraint is finally brought to the form

b∗L−a∗U ≤ b+a−a∗b−1. The method ensures that this new constraint is consistent

in this post-condition:

ensures forall val :: EvalInequality(darkIneq, val) <==>
InverseInequalityStarStar(

lowerIneq.left, upperIneq.right,
lowerIneq.right.terms[0].coeff,
upperIneq.left.terms[0].coeff, val)

Code snippet 3.8: BuildGeneratedPair post-condition

where InverseInequalityStarStar is a predicate that checks if aU − bL > ab− a− b:

subtract(multiply(a, EvalExpr(U,val)),
multiply(b, EvalExpr(L,val))).val >

subtract(subtract(multiply(a, b), a), b).val.

PartitionDark applies this elimination step for each pair in the system and re-

turns a DarkTraceSystem that is used later by a series of functions that retrieve the

inequalities from the fields base and every p.c from pairs. Now, we can state: if proj

SAT =⇒ original SAT.

An example that demonstrates how transformations apply to a system is:
2y + 1 ≤ 2x,

3x ≤ y + 10,

−y ≤ −2


dark eliminate x−−−−−−−−→

 4y ≤ 15,

−y ≤ −2

 dark eliminate y−−−−−−−−→ {0 ≤ 7}

3.2.2 Correctness Proof

The DarkShadow method only proves satisfiability in one direction:

exists v :: SystemHolds(darkStep, v)
==>
exists w :: SystemHolds(sys, w).

Code snippet 3.9: DarkShadow satisfiability implication

We start from SplitSystemForDark, which constructs an object of the type Parti-

tionedSystem and proves that every inequality in the original system appears in one

of the three fields of the partitioned one, as shown in the code snippet 3.9.

Next, we use this partitioned system to build the new constraints by calling Ex-

istsXInTrace → PartitionDark → BuildGeneratedPair.

The most important post-condition of BuildGeneratedPair shows that the new

constraint is consistent (according to the code snippet 3.8). Because the method splits

the construction of the new inequality in small parts like L,U, a, b, it can use a lemma

to transform a syntactic construction into a semantic one. The lemma DarkIneqEquiv-

alentToInverseStarStar proves how the new constraint built out of the pair is actually

the condition that the Dark Shadow needs to prove satisfiability.

Back in PartitionDark, we prove most of the conditions necessary for the final

implication. Because this is the method that builds a DarkTraceSytem, its invariants

have to prove existence and keep track of properties. The exterior loop, which iterates

through ps.pos, fixes an inferior limit, and the exterior one covers the superior ones.

We also use an auxiliary variable pairIdx to store the position in trace.pairs where

the new pair was added. This way, after constructing a new object PairDark, the proof

can refer directly to the new inserted element (trace.pairs[pairIdx]) and can

immediately verify its properties: that it was generated from the current bounds, that

it has the correct dark form and that it is associated to the pair of indexes (j, i).

pairIdx := pairIdx[(iu, il) := k];
assert (iu, il) in pairIdx;
assert pairIdx[(iu, il)] == k;
assert generatedPairs[pairIdx[(iu, il)]].a == ps.pos[iu];
assert generatedPairs[pairIdx[(iu, il)]].b == ps.neg[il];

After keeping track of those properties regarding pairIdx using invariants, after

the loops, we extract the properties to connect the trace to the partitioned system:

forall j, i | 0 <= j < |ps.neg| && 0 <= i < |ps.pos|
ensures exists p {:trigger trig(i, j, p)} ::
trig(i, j, p) && p in trace.pairs
&& p.b == ps.neg[j] && p.a == ps.pos[i]

{
assert (i, j) in pairIdx;
var k := pairIdx[(i, j)];
assert 0 <= k < |generatedPairs|;
var p := generatedPairs[k];
assert p.a == ps.pos[i];
assert p.b == ps.neg[j];
assert trace.pairs == generatedPairs;
assert p in trace.pairs;
assert trig(i, j, p) && p in trace.pairs

&& p.b == ps.neg[j] && p.a == ps.pos[i];
}

The :trigger notations are necessary for instantiating the quantification used

by the SMT solver in Dafny. The expression trig(i,j,p) is used because the witness

p needs to be correlated with two indices of the cartesian product, making it difficult

for the solver to verify the condition.

After proving that the inequalities from trace come from the partitioned system,

the method also proves a few predicates starting from the post-conditions of Build-

GeneratedPair (shown in code snippet 3.8).

ensures forall p :: p in trace.pairs ==>
PairDarkHasWitnessForX(p, x)

ensures forall p :: p in trace.pairs ==>
PairDarkIsStarStarForX(p, x)

ensures forall p :: p in trace.pairs ==>
PairDarkIsStarForXRewritten(p, x)

Code snippet 3.10: Post-conditions of PartitionDark

Those predicates take the property of a dark inequality being constructed cor-

rectly and add conditions for validity and structure constraints. PairDarkHasWitness-

ForX says if the evaluation of the remaining variables satisfies the dark constraint p.c,

then there exists an integer value for x that simultaneously satisfies the two original

constraints p.a and p.b. This method still does not lift the proof of satisfiability to the

whole system, it only states it for each pair of bounds.

The method ExistsXInTrace is the method used as the main proof in Dark Shadow.

This is only an intermediate method that calls PartitionDark and obtains the trace,

along with its properties, and lifts them from a local satisfiability proof over pairs, to a

global one, over the whole system. A very important lemma for this step is ExistsX. A

simple schema the lemma follows in its proof is:

TraceBaseAndPairsCHold(trace, val)
↓

forall p in trace.pairs: EvalInequality(p.c, val)
↓

// through StarStar:
forall lower, upper:

the interval between a lower and an upper bound contains
an integer value

↓
DarkConstraintsGiveCommonInteger

↓
exists n:int:

n satisfies all lower bounds and all upper bounds
↓

ps.zero does not depend on x
↓

UpdateValuation(val, x, n) satisfies ps.pos, ps.neg, ps.zero

The predicate UpdateValuation is used specifically to express the new value for

a variable in the context of an existing valuation for the rest of the variables.

Using this lemma, ExistsXInTrace can finally prove its post-conditions:

ensures forall j, i | 0 <= j < |ps.neg|
&& 0 <= i < |ps.pos| ::

exists p :: p in trace.pairs && p.b == ps.neg[j]
&& p.a == ps.pos[i]

ensures forall p {:trigger p in trace.pairs} :: p in
trace.pairs ==> PairDarkTracksSplit(p, ps)

ensures forall v0 :: (TraceBaseAndPairsCHold(trace, v0)
&& |ps.pos| > 0 && |ps.neg| > 0
==> exists n:int ::

(forall p :: p in ps.pos ==> EvalInequality(p,
UpdateValuation(v0, x, n)))

&&
(forall p :: p in ps.neg ==> EvalInequality(p,

UpdateValuation(v0, x, n)))
&&
(forall p :: p in ps.zero ==> EvalInequality(p,

UpdateValuation(v0, x, n))))

The first one states that all inequalities from the partitioned system are also in

the trace. The second post-condition expresses that all inequalities in trace are also

in the partitioned system. The last one is the one that will be used to prove the final

satisfiability property.

DarkEliminateStepTrace combines the local properties established during par-

titioning and pair construction in order to prove the final soundness property of one

dark-elimination step. The proof relies on three main facts:

• the partitioned representation is equivalent to the original system;

• every inequality from the original system is preserved in one of the partitions;

• if the generated dark-shadow constraints hold under a valuation, then there ex-

ists an integer value n for the eliminated variable x such that the partitioned

system holds under the updated valuation.

Together, these facts allow the method to lift the witness property from the partitioned

system back to the original system:

ensures forall v :: (SystemHolds(TraceToIneqSystem(trace), v)
==> exists n:int :: SystemHolds(sys,

UpdateValuation(v, x, n)))

3.3 Grey Shadow

If the Dark Shadow is unsatisfiable and the Real Shadow is satisfiable, we cannot

yet establish the satisfiability of the original system. Therefore, the Omega Test uses

the Grey Shadow to form a finite number of subproblems that search for solutions

in the intervals missed by the Dark projection. If any of the grey subproblems are

satisfiable, then the original system is also satisfiable. If there is no satisfiable grey

subproblem, then the original system is unsatisfiable.

3.3.1 Building the Grey subproblems

In the Grey Shadow, a pair of the form L ≤ ax, bx ≤ U is considered offending if

both of those conditions are true:

• the real interval is non-empty: bL ≤ abx ≤ aU ;

• the dark condition does not hold: (aU − bL > ab− a− b) ≡ (aU − bL ≤ ab− a− b).

To be able to verify this at all steps of the elimination, the OmegaTest (presented

in figure 3.1) is built around the Dark Shadow. If the dark projection is proven unsatis-

fiable at any time during the elimination algorithm, then the grey branches are created

to complete the search for solutions.

The grey section of the OmegaTest consists of two while loops:

while (j < |ps.neg| && !anySubproblemSat)
// 0 ≤ i ≤ ab−a−b

b
while (i <= maxOffset && !anySubproblemSat)

// build eq := ax = L+ i
// sysReplaced := Sys ∪ eq
var greyValid := OmegaTest(sysReplaced, elimOrder);

A grey equality is built and added to the system using the method Substitute-

UnitEqualityInSystem. Since this system implementation cannot handle equations

(only inequalities), the method adds the equation ax = L+ i as ax ≤ L+ i, L+ i ≤ ax.

Before the recursive call on the new system, the algorithm pre-processes the equa-

tion using two methods:

• CheckGreyBranchGcd: it computes the GCD of all coefficients of the variables

and checks if it divides the equality constant. If it doesn’t, then it means this

equation is unsatisfiable for integer values and therefore the branch is also unsat-

isfiable;

• SubstituteUnitEqualityInSystem: if the coefficient of x in the lower bound is 1

(a == 1 and ax = L + i ⇐⇒ x = L + i. This means that we can replace

any appearance of x in the system with L + i and there is no need for adding

ax ≤ L+ i, L+ i ≤ ax as additional constraints.

3.3.2 Verification of Grey Branching

Each grey branch created for an equation is tested using a recursive call to OmegaT-

est, except for the case in which the equation is unsatisfiable. If one branch holds under

a valuation, we can state that the original system also holds.

assert exists v :: SystemHolds(substituted, v);
ghost var v :| SystemHolds(substituted, v);
assert exists w: Valuation :: SystemHolds(sys, w);

Proving this is straightforward: since substituted is just sys ∪ eq, then sys ⊆

substituted which means that substituted SAT =⇒ sys SAT.

However, if, after exhausting every grey branch, the algorithm does not find any

satisfiable subproblem, then it states that the original system is unsatisfiable. This proof

is a contradiction argument.

We assume the algorithm says all grey branches are UNSAT, but the original sys-

tem sys still has a model v. This means that the partitioned system also holds:

assert PartitionedSystemHolds(ps, v);

For a system to have a solution, it means that either the Dark Shadow is satisfi-

able, or the Grey Shadow. This is proved by the lemma OriginalModelImpliesDark-

OrGreyEquality:

ensures SystemHolds(TraceToIneqSystem(trace), val)
||

exists p, k: int ::
p in trace.pairs &&
0 <= k &&
k as real <=

GreyShadowUpperBound2(p.b.right.terms[0].coeff,

p.a.left.terms[0].coeff).val
&& GreyEquality(p.b.left, k,

p.b.right.terms[0].coeff, x, val)

We also state that the dark projection does not hold and there must exist a pair of

bounds that do not respect the dark condition for satisfiability (we know this because

we are on the else branch of the Dark Shadow): L ≤ a ∗ x ≤ L + greyBound . This is

equivalent to ax = L + i, where i ∈ [0, greyBound]. The proof then shows that this

i is one of the offsets the algorithm enumerated, using GreyRealBoundImpliesInte-

gerOffsetBound.

Since the algorithm proved every such grey branch UNSAT, this particular branch

must be UNSAT. But the original model v satisfies sys , and it also satisfies a∗x = L+ i,

so v satisfies that grey branch. This is shown by GreyEqualityImpliesSourceSys-

temHolds.

Therefore, we have a contradiction: branch is UNSAT, but v is a model of branch.

So the original assumption was false. Therefore, if Dark Shadow is UNSAT and every

required grey branch is UNSAT, then the original system is UNSAT.

3.4 Final proof of the Elimination Algorithm

The OmegaTest is a variable elimination procedure. At each step, it tries to re-

move a variable until the system is variable-free. At this point, the system contains

only Inequality objects of the form 2 ≤ 3, 5 ≤ 2 and can be checked using the method

CheckProjectionNoVars.

If all the constant inequalities are satisfied, then it returns True and using the

lemma VariableFreeSatisfiableImpliesModel, it proves that exists val: Valuation

:: SystemHolds(sys, val). This lemma uses a ghost Valuation variable (ghost

var zeroVal: Valuation := => 0), which maps every variable name to 0, to

show that there exists a Valuation for which the system is satisfiable.

If there exists one or more inequalities that are not satisfied, then it can ensure

that forall v :: !SystemHolds(sys, v).

After the base case, the method handles the recursive elimination step. The first

variable from elimOrder is selected: x := elimOrder[0] and the remaining vari-

ables form nextElimOrder.

method OmegaTest(sys: IneqSystem, elimOrder: seq<string>)
returns (sat: bool)

requires ValidSystem(sys)
requires IntegerCoeffSys(sys)
requires forall x :: TermInSystem(sys, x) ==> x in

elimOrder
requires forall i, j :: 0 <= i < j < |elimOrder| ==>

elimOrder[i] != elimOrder[j]
ensures sat ==> (exists v :: SystemHolds(sys, v))
ensures !sat ==> (forall v :: !SystemHolds(sys, v))
decreases *

Code snippet 3.11: The definition of the OmegaTest

The method receives a system that is normalized and has only terms with integer

coefficients. Also, the elimOrder needs to contain all unique variables from the system.

The method ensures that the value returned is consistent with the satisfiability of the

system.

The method first calls the Dark Shadow elimination steps, then it establishes that

the dark projection is a valid recursive problem. The method proves that the projected

system contains only variables from nextElimOrder, because x has been eliminated.

It also proves that the trace correctly relates the generated constraints to the original

partition: darkTrace.base == ps.zero, every pair in the trace satisfies PairDark-

TracksSplit, and every generated constraint satisfies the dark-star condition for x (as

shown in the code snippet 3.2.2).

If this recursive call returns True, then by the postcondition of OmegaTest and

of DarkEliminateStepTrace, there exists a valuation satisfying the dark projection and

therefore the original system also.

If the dark projection is unsatisfiable, the method constructs the Real Shadow.

The proof establishes that this projection is still a valid recursive problem and that it

contains only variables from nextElimOrder. If the real shadow has no model, then

the original system cannot have a model either. In the code, this is expressed by the

implication:

forall v :: !SystemHolds(realS, v))
==>
forall v :: !SystemHolds(normReal, v)).

Together with the equivalence between the normalized system and the real pro-

jection, this allows the method to derive forall v :: !SystemHolds(sys, v)

and to return False.

The last case is when the dark shadow is unsatisfiable, but the real shadow is

satisfiable. In this situation, the algorithm cannot immediately return either True or

False, so it explores the Grey Shadow branches. Since the added equality only restricts

the original system, any valuation that satisfies a grey branch also satisfies the original

system. Therefore, if one recursive call on a grey branch returns True, then OmegaTest

can also return True for the original system.

If all grey branches return False, the method keeps track of the unsatisfiability

information and then proves forall v :: !SystemHolds(sys, v) and returns False.

The post-conditions of OmegaTest relate the satisfiability of the system to the

value returned:

ensures sat ==> (exists v :: SystemHolds(sys, v))
ensures !sat ==> (forall v :: !SystemHolds(sys, v))

Chapter 4

Termination in the OmegaTest

By applying the version of OmegaTest described in the manuscript [1], without

additional preprocessing of the equations generated in the Grey Shadow, we encoun-

tered a potential termination problem. The termination of the Real Shadow and the

Dark Shadow can be proved from the way those projections are built: at each step, we

remove one variable and the method terminates when elimOrder is empty.

The Grey Shadow, however, does not eliminate the current variable being pro-

cessed. It just adds one more constraint to the system, the equation ax = L + i, not

eliminating x; instead, the search space is restricted to one possible value of ax near

the lower bound L. Since we represent the equation as ax ≤ L + i, L + i ≤ ax, if

those inequalities end up generating grey equations of their own, the next recursive

call might be done on the system: sys ∪ ax ≤ L + i ∪ L + i ≤ ax ∪ ax ≤ L + i ∪

L + i ≤ ax. The next call will generate two more such inequalities by starting from

two of those bounds and so on, causing the method to never reach a conclusion and

terminate.

An example would be the system 2x ≤ 1,−2x ≤ −1. OmegaTest firstly computes

the dark projection and notices it is unsatisfiable; therefore, it also generates the real

projection. Since the real projection is satisfiable, the algorithm also constructs the grey

branches. In this case, there is only one lower bound (1 ≤ 2x) and only one value for

i ∈ [0, 2∗2−2−2
2

] =⇒ i = 0. The only equation it constructs is 2x = 1, which, added to

the system, results in the same inequalities we started from.

 2x ≤ 1,

−2x ≤ −1

 Dark Shadow−−−−−−−→ {0 ≤ −1} Real Shadow−−−−−−−→
{
1

2
≤ x ≤ 1

2

}

Grey Shadow−−−−−−−→



2x ≤ 1,

−2x ≤ −1,

2x ≤ 1,

−2x ≤ −1


.

This example revealed a potential conceptual problem in the manuscript used as

the basis for the formalization of this implementation. The grey-shadow step strength-

ens the system by adding an equation, but it does not necessarily eliminate the current

variable or decrease the elimination order. Therefore, the termination argument used

for the Real Shadow and the Dark Shadow (decreases |elimOrder|) cannot be

applied directly to the Grey Shadow. In particular, when the generated equation is

represented again as two inequalities, the recursive call may reconstruct the same kind

of grey branch and fail to make progress.

To avoid this issue in practice, we added supplementary equation-preprocessing

steps, similar to the equality-handling techniques described by William Pugh [4].

For example, 2x = 4y − 6z + 1 can never be satisfied for any integer valuation of

x, y, z. The GCD is gcd(2, 4, 6) = 2 and 2 ∤ 1, therefore the equation has no integer solu-

tion. We can then conclude the system is unsatisfiable without recalling OmegaTest to

check it.

Another method of pre-processing equations is in the case of a unit coefficient for

x. If a = 1, then ax = L + 1 is just x = L + i, so x has integer solutions if the variables

in L also have integer solutions. We replace every occurrence of x in the system with

L+ i and can now recall OmegaTest(substituted, nextElimOrder).

Even with these equation processing methods, there are still cases that fall outside

of those situations and need to be verified with the OmegaTest, causing an infinite re-

cursion. For this reason, the termination of the full algorithm was not formally proved

in this thesis.

The implementation includes a heuristic that consists of pre-processing the equa-

tions generated by the Grey Shadow before making the recursive call. This avoids the

infinite loop in some cases, but it is not formally verified as a complete termination

proof.

Therefore, the formal result established here represents partial correctness: if the

algorithm terminates, then the result it returns is the expected one. Proving total cor-

rectness, including termination of all grey-shadow branches, remains future work.

Chapter 5

Future Improvements

The current implementations has a few limitations concerning efficiency, some

about the nature of the algorithm and others about the total verification cost in Dafny.

5.1 Efficiency Considerations

The algorithm is very sensitive to the number of inequalities and variables in a

system. At each step of eliminating variables, the algorithm does a cartesian product

over the set of lower bounds and the set of upper bounds, so the number of inequalities

can grow very fast. The most complex case is when we reach the Grey Shadow and

have to build a new system for each equality created, possibly creating many subprob-

lems. According to the lecture slides taken as reference, Omega Test becomes very

sensitive to coefficients and big coefficients lead to many grey subproblems [1].

Another important limitation is the representation of the rational values in the

implementation. The type Rational stores the numerator and the denominator, but the

operations are not simplified by dividing them to the greatest common denominator.

For example, after repeated operations like addition, subtraction or multiplication, we

might end up with different representations of the same values, like 1
2

and 2
4
. For

Dafny, the issues is not only about the numerical dimension, but also about proving

equivalence using num, den and val, which implied many auxiliary lemmas.

A problem specific to the verification of Dafny is the use of

{:vcs split on every assert}, which indicates the Z3 verifier to split the proof

into smaller obligations, instead of a big VC that times out and cannot be proven.

Although this helps Dafny verify the definition, it slows down the verification process,

making it more costly.

Finally, a future version of this algorithm could be optimized by implementing a

simplification of fractions using the greatest common denominator, eliminating redun-

dant constraints and maintaining coefficients as small as possible.

5.2 Handling Equality Constraints

One problem discussed in Chapter 5 is that in the Grey Shadow, the generated

subproblem contains an equality of the form ax = L + i. If this equality is encoded

only as two inequalities ax ≤ L+ i, L+ i ≤ ax, the recursive call receives a system that

might induce an infinite recursion. The variable x is not eliminated, so the algorithm

could fail to make progress.

A solution for handling the non-termination issue regarding the OmegaTest is

to encode the equations from the Grey Shadows as divisibility conditions. This so-

lution is presented in the original Omega Test paper [4]. An equality ax = L + i can

only have a solution if a | (L + i). If this condition is not satisfied, the branch can be

discarded immediately. Otherwise, the equality can be used to simplify the system,

reduce coefficients, or eliminate a variable.

For a single linear equality over unrestricted integer variables, a1x1+ · · ·+anxn =

c, the divisibility condition gcd(a1, . . . , an) | c characterizes whether the equality itself

has an integer solution. However, when the equality appears inside a larger system

of constraints, this condition is not sufficient for the whole system to be satisfiable. It

only shows that the equality is not contradictory by itself; the remaining inequalities

must still be checked.

Therefore, a solution is to encode those divisibility conditions in the system and

implement methods of verifying them when handling equivalence, evaluation or sat-

isfiability.

This divisibility reasoning is important for grey branches, because it turns the

added equality into an arithmetic constraint that can be checked directly, rather than

leaving it as another pair of inequalities. In the current implementation, this part is

only partially handled: the equality is added to the system, but a complete version

would also use divisibility constraints to guarantee that recursive calls make progress.

Chapter 6

AI Usage

This chapter briefly describes the role of AI-assisted tools in the project. Since

the thesis involves both implementation and formal verification, these tools helped

generate examples, explain errors, explore proof ideas, and identify possible reasons

for Dafny verification failures. They were useful when trying to understand why a

certain assertion did not follow automatically or why a mathematical argument im-

plemented in Dafny resulted in a verification error. For example, in the proofs about

rational arithmetic and sums of sequences, the AI agents helped by explaining how

calc blocks could be used to guide Dafny through algebraic transformations.

However, in most cases, the suggested proof structure was too general and it

missed proof steps that caused Dafny to fail to verify mathematical properties. For

example, it did not manage to find the verifying issue and solution for this prompt:

Why does the first assertion holds but the second doesn’t:

assert SumTerms(terms Lb, val).val ==

multiply(SumTerms( L.terms, val), b).val;

assert SumTerms(terms Lb, val).val ==

SumTerms(L.terms, val).val * b.val;

The agent responded to replace the assertions with a calc block. However, when

the calc block was implemented, Dafny failed to verify it because of a verifier timeout.

The issue was only solved after implementing a lemma to prove the intermediate step.

The most useful aspect of using this agent was to understand concepts regarding

the small proofs from the Omega Test. For example, it gave a correct mathematical

answer for the prompt:

Can you give me an example for fourier-motzkin that has integer solutions, but

it can’t find them using dark shadow?

The answer given was the following system that does have integer solutions, but

not ones discovered by the Dark Shadow:

2x ≤ y + 1,

y ≤ 2x− 1,

x ≤ 5,

x ≥ 0,

y ≥ 0

In conclusion, the AI conversational tools represented a small part of the im-

plementation, used mostly to fix local verification errors. Each answer was carefully

checked before being introduced in the code.

Conclusions

The goal of this thesis was the implementation, specification and verification of

the Omega Test algorithm in Dafny. The Omega Test is a variation of the Fourier-

Motzkin elimination algorithm, extended to find whether a system of inequalities has

an integer solution.

The implementation started from the representation of rational numbers, terms,

inequalities and different structures that represent systems. For each defined structure,

a set of predicates and functions were introduced in order to transform, evaluate and

compare the systems. The central part of this work is represented by the implementa-

tion and specification of the Real Shadow, Dark Shadow and the Grey Shadow.

To build one of those projections, each construction step needed precise specifi-

cations. The implementation showed that each generated system remained valid, that

the Real Shadow is sound with respect to integer solutions, that the Dark Shadow is

sufficient to decide satisfiability over integers and that the Grey Shadow provides the

completeness of the algorithm. The result is accumulated from different branches of

recursion and related at the end to the proved property.

As future work, the implementation could be improved by reducing large co-

efficients during arithmetic operations. This would help reduce the number of new

constraints generated in the Grey Shadow and could reduce the complexity of the

algorithm. Another important improvement, regarding the termination problem in

some Grey Shadow branches, would be to implement a different termination heuristic

or to handle the equations generated by the grey branches in a way that eliminates the

variable they constrain.

In conclusion, this implementation can be used to construct both the real projec-

tion and the integer projections of a system of linear constraints. It can also be used

to reason about the satisfiability of those systems and to show that satisfiability over

integer variables is a decidable problem.

